An Alexander-Determinant-PolyTime Puzzle

Pensieve Header: An Alexander-Determinant-PolyTime Puzzle, based on Projects/WKO/wA.nb

```
<< KnotTheory`
```

```
Loading KnotTheory` version of April 20, 2009, 14:18:34.482.
Read more at http://katlas.org/wiki/KnotTheory.
```

The Program and a Test Run on Knot[8,17]

```
K = Knot[8, 17];
Alexander[K][X]
```

$$11 - \frac{1}{X^3} + \frac{4}{X^2} - \frac{8}{X} - 8 X + 4 X^2 - X^3$$

Generating Gauss Codes

```
GC[K_] := GC @@ (
   PD[K] /. X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]],
      Ar[l, i, +1], Ar[j, i, -1]
     ]
  )
```

```
GC[K]
```

```
GC[Ar[1, 6, 1], Ar[7, 14, 1], Ar[3, 8, -1], Ar[13, 2, -1],
 Ar[5, 12, -1], Ar[9, 4, -1], Ar[11, 16, 1], Ar[15, 10, 1]]
```

Generating the Trapping Matrix

```
Tij[Ar[ti_, hi_, si_], Ar[tj_, hj_, sj_]] := If[
   ti < hj < hi || hi < hj < ti,
   1, 0
  ];
T[K_] := Module[
  {gc = GC[K]},
  Outer[Tij, List @@ gc, List @@ gc]
 ]
```

```
T[K] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 1 & 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 & 1 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 1 & 0 & 1 & 0 & 1 & 1 & 0 & 1 \\ 1 & 0 & 1 & 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 1 & 0 & 0 & 0 \end{pmatrix}$$

The Diagonal Matrix of Signs S

```
Si[Ar[ti_, hi_, si_]] := Sign[hi - ti] * si;
S[K_] := DiagonalMatrix[Si /@ (List @@ GC[K])];
S0[K_] := MatrixExp[-Log[X] S[K]];
S1[K_] := S0[K] - IdentityMatrix[Crossings[K]]
```

```
MatrixForm /@ {S[K], S0[K], S1[K]}
```

$$\left\{\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} \frac{1}{X} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{X} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & X & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{X} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & X & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{1}{X} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{X} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & X \end{pmatrix}, \begin{pmatrix} -1+\frac{1}{X} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1+\frac{1}{X} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -1+X & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1+\frac{1}{X} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1+X & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{X} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{X} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+X \end{pmatrix}\right\}$$

wA

```
M[K_] := IdentityMatrix[Crossings[K]] - T[K].S1[K];
M[K] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 1-\frac{1}{X} & 0 & 1-\frac{1}{X} & 0 & 0 \\ 0 & 1 & 1-X & 0 & 1-X & 0 & 0 & 1-X \\ 1-\frac{1}{X} & 0 & 1 & 0 & 0 & 1-\frac{1}{X} & 0 & 0 \\ 1-\frac{1}{X} & 0 & 1-X & 1 & 1-X & 1-\frac{1}{X} & 0 & 1-X \\ 1-\frac{1}{X} & 0 & 1-X & 0 & 1 & 0 & 0 & 1-X \\ 1-\frac{1}{X} & 0 & 1-X & 0 & 0 & 1 & 0 & 0 \\ 0 & 1-\frac{1}{X} & 0 & 0 & 1-X & 0 & 1 & 0 \\ 0 & 1-\frac{1}{X} & 0 & 0 & 1-X & 0 & 0 & 1 \end{pmatrix}$$

```
wA[K_] := Det[M[K]];
wA[K]
```

$$11-\frac{1}{X^3}+\frac{4}{X^2}-\frac{8}{X}-8X+4X^2-X^3$$

Free Play

```
MatrixForm[
 S0[K] / X /. X → 1 / Sqrt[X]
]
```

$$\begin{pmatrix} X & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & X & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & X & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & X & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & X & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

```
MatrixForm[Expand[
   M[K].(S0[K] / X /. X → 1 / Sqrt[X])
  ] /. X → X + 1]
```

$$\begin{pmatrix} 1+X & 0 & 0 & X & 0 & X & 0 & 0 \\ 0 & 1+X & -X & 0 & -X & 0 & 0 & -X \\ X & 0 & 1 & 0 & 0 & X & 0 & 0 \\ X & 0 & -X & 1+X & -X & X & 0 & -X \\ X & 0 & -X & 0 & 1 & 0 & 0 & -X \\ X & 0 & -X & 0 & 0 & 1+X & 0 & 0 \\ 0 & X & 0 & 0 & -X & 0 & 1+X & 0 \\ 0 & X & 0 & 0 & -X & 0 & 0 & 1 \end{pmatrix}$$